Retour au site web

Exercice 2-2 : Polynômes de Hermite

a) Appliquons la relation de récurrence donnée dans le texte, en partant des polynômes $H_0 = 1$ et $H_1 = 2x$:

$$\begin{array}{llcl}
(n=1) & H_2 & = & 2xH_1 - 2\times H_0 = 4x^2 - 2; \\
(n=2) & H_3 & = & 2xH_2 - 2\times 2H_1 = 8x^3 - 12x; \\
(n=4) & H_4 & = & 2xH_3 - 2\times 3H_2 = 16x^4 - 24x^2 - 6(4x^2-2) \\
 & & = & 16x^4 - 48x^2 + 12.
\end{array}$$

Vous constatez que les coefficients des polynômes successifs sont rapidement croissants avec n.

b) La relation de récurrence s'applique tout autant pour une valeur particulière de x, par exemple $x = 0{,}5$. Ainsi,

$$\begin{array}{lcl}
H_0(0{,}5) & = & 1; \\
H_1(0{,}5) & = & 1; \\
H_2(0{,}5) & = & 2\times 0{,}5\times 1 - 2\times 1 = -1; \\
H_3(0{,}5) & = & 2\times 0{,}5(-1) - 2\times 2\times 1 = -5; \\
H_4(0{,}5) & = & 2\times 0{,}5(-5) - 6(-1) = 1; \\
H_5(0{,}5) & = & 2\times 0{,}5\times 1 - 8(-5) = 41.
\end{array}$$

c) Le programme qui suit permet le calcul de $H_n(x)$. Nous avons choisi d'afficher H_2, H_3, H_4 et H_5, qui ont des valeurs pas trop disparates. La croissance rapide des valeurs numériques avec l'indice est gênante pour certaines applications ; certains auteurs adoptent une relation de récurrence un peu différente :

$$H_{n+1}(x) = xH_n(x) - nH_{n-1}(x),$$

qui fournit des polynômes proportionnels aux précédents mais dont les valeurs croissent lentement avec n (voir : Christian Guilpin : *Manuel de calcul numérique appliqué*, ch. 10, EDP Sciences, Les Ulis, 1999).

```
//Exercice 2-2: polynômes de Hermite par la relation
//de récurrence H_{n+1} = 2xH_n - 2nH_{n-1}
function fn = H(n,x)
if n == 0, fn = 1, end
if n == 1, fn = 2*x, end
if n > 1
  k = 1; Havder = 1; Hder = 2*x;
  while k < n
    H = 2*x.*Hder - 2*k*Havder;
    Havder = Hder;
    Hder = H;
    k = k+1;
  end
  fn = H;
end
endfunction
xmin = -3; xmax = 3;
x = xmin:0.03:xmax;
```

```
clf();
scf(0); ha = gca();
plot2d(x,[H(2,x)',H(3,x)',H(4,x)',H(5,x)'],..
    rect=[xmin,-150,xmax,150]);
xsegs([xmin,xmax],[0,0])
ha.x_label.font_size = 4;ha.y_label.font_size = 4;
xlabel("$x$"); ylabel("$H_n(x)$");
xtitle("polynomes de Hermite, n = 2,3,4,5 ")
```